Inverter

Write the vhdl for the circuit. Here we consider the example of an inverter

inverter.vhd

library IEEE;											        use IEEE.std_logic_1164.all;
entity inverter is								                                    port (input    : in std_logic;										            	output   : out std_logic);								         end inverter;
architecture structural of inverter is								     begin														  output <= not (input);    					                           	        end structural;
Run it through design vision and we can save a .v(verilog) file 
inverter.v
module inverter ( \input , \output  );									  input \input ;											  output \output ;
 INV_X1 U2 ( .A(\input ), .ZN(\output ) );
Endmodule

Add the design(.v netlist) in encounter, Place and Route it and do RC Extraction where we get a .spef file. 
A typical SPEF file will have 4 main sections
– a header section,
– a name map section,
– a top level port section and
– the main parasitic description section.
Generally, SPEF keywords are preceded with a *. For example, *R_UNIT, *NAME_MAP and *D_NET.
inverter.spef

*SPEF "IEEE 1481-1998"
*DESIGN "inverter"
*DATE "Sat Mar 19 16:00:57 2011"
*VENDOR "Silicon Perspective, A Cadence Company"
*PROGRAM "Encounter"
*VERSION "10.10-s002_1"
*DESIGN_FLOW "PIN_CAP NONE" "NAME_SCOPE LOCAL"
*DIVIDER /
*DELIMITER :
*BUS_DELIMITER []
*T_UNIT 1 NS
*C_UNIT 1 PF
*R_UNIT 1 OHM
*L_UNIT 1 HENRY

*Header Information
*The header section is 14 lines containing information about
* the design name,
*the parasitic extraction tool,
* naming styles
* and units.
*When reading SPEF, it is important to check the header for units as they vary across tools. By *default, SPEF from Astro will be in pF and kOhm while SPEF from Star-RCXT will be in fF and *Ohm.

*Name Map Section
*To reduce file size, SPEF allows long names to be mapped to shorter numbers preceded by a *. *This mapping is defined in the name *map section.

*NAME_MAP

*1 input
*2 output
*3 U2

*Port Section
*The port section is simply a list of the top level ports in a design. They are also annotated as *input, output or bidirect with *an I, O or B.

*PORTS

*1 I *C 11 6
*2 O *C 11 6


*Parasitics
*Each extracted net will have a *D_NET section. This will usually consist of a *D_NET line, a *CONN section, a *CAP section, *RES *section and a *END line. Single pin nets will not have a *RES section. Nets connected by abutting pins will not have a *CAP *section.

*The *D_NET line tells the net name and the net's total capacitance. This capacitance will be *the sum of all the capacitances in the *CAP section.

*D_NET *1 0.000334148


*The *CONN section lists the pins connected to the net. A connection to a cell instance starts with a *I. A connection to a top *level port starts with a *P.
*The syntax of the *CONN entries is:
*I <pin name> <direction> *C <xy coordinate> <loading or driving information>
*Where:
*  The pin name is the name of the pin.
*  The direction will be I, O or B for input, output or bidirect.
*  The xy coordinate will be the location of the pin in the layout.
*  For an input, the loading information will be *L and the pin's capacitance.
*  For an output, the driving information will be *D and the driving cell's type.
*  Coordinates for *P port entries may not be accurate because some extraction tools look for the physical location of the logical *port (which does not exist) rather then the location of the corresponding pin.


*CONN
*P *1 I *C 11 6 *L 0
*I *3:A I *C 7 6 *L 0 *D INV_X1

**CAP Section
*The *CAP section provides detailed capacitance information for the net. Entries in the *CAP section come in two forms, one for a *capacitor lumped to ground and one for a coupled capacitor.
*A capacitor lumped to ground has three fields,
*  an identifying integer,
*  a node name and
*  the capacitance value of this node
*  e.g
*o1 regcontrol_top/GRC/U9743:E 0.936057
*A coupling capacitor has four fields,
*  an identifying integer,
*  two node names and
*  The values of the coupling capacitor between these two nodes
*  E.g
*o2 regcontrol_top/GRC/U9409:A regcontrol_top/GRC/U10716:Z 0.622675
*If netA is coupled to netB, the coupling capacitor will be listed in each net's *CAP section.


*CAP
1 *1 0.000167074
2 *1:1 0.000167074


**RES Section
*The *RES section provides the resistance network for the net.
*Entries in *RES section contain 4 fields,
*  an identifying integer,
*  two node names and
*  the resistance between these two nodes.
*  E.g
*o1 regcontrol_top/GRC/U9743:E regcontrol_top/GRC/U9407:Z 10.7916
*The resistance network for a net can be very complex. SPEF can contain resistor loops or *seemingly ridiculously huge resistors even if the layout is a simple point to point route. This is *due how the extraction tool cuts nets into tiny pieces for extraction and then mathematically *stitches them back together when writing SPEF.


*RES
1 *1:2 *3:A 6
2 *1:1 *1:2 5
3 *1 *1:1 14.5804
*END

*D_NET *2 0.000331591

*CONN
*I *3:ZN O *C 7 5 *L 0 *D INV_X1
*P *2 O *C 11 6 *L 0

*CAP
1 *3:ZN 8.42534e-05
2 *2:1 8.42534e-05
3 *2:3 8.15419e-05
4 *2 8.15419e-05

*RES
1 *2:3 *2 7.11607
2 *2:2 *2:3 5
3 *2:1 *2:2 6
4 *3:ZN *2:1 12.2185
*END


Using the parasitic capacitance and resistance from the above .spef file the inverter circuit can be drawn as follows
[image: ]
With this as the basis, rename the nodes with simple numbers, write a design in hspice- .sp file 
[image: ]
inverter netlist
vdd 31 0 dc 5v
*DC_value   PEAK_value  delay    risetime   falltime   pulse_width   period
Vin 1 0 pulse(5 1m 0 1n 1n 50u 100u)
v1 31 3 dc 0v
v2 32 0 dc 0v
r1  1 2 14.58
c1  1 0 0.000161pF
c2  2 0 0.000161pF
r4  5 7 5
r11 2 21 5
r12 21 22 6

c5  7 0 0.0000815pF
r5  7 8 12.2185
c6  8 0 0.0000815pF
XINV 22 5 3 32 INVERT
*
.MODEL CMOSN NMOS (  
+VERSION = 3.1           LEVEL=49         TNOM    = 27             TOX     = 3.1E-9
+XJ      = 1E-7           NCH     = 2.3549E17      VTH0    = 0.0485675
+K1      = 0.3634695      K2      = -0.0277136     K3      = 1E-3
+K3B     = 3.9409342      W0      = 1E-7           NLX     = 8.900764E-7
+DVT0W   = 0              DVT1W   = 0              DVT2W   = 0
+DVT0    = 1.2979753      DVT1    = 0.1605369      DVT2    = 0.2509178
+U0      = 436.2279588    UA      = -3.51975E-10   UB      = 3.216114E-18
+UC      = 4.919099E-10   VSAT    = 1.930256E5     A0      = 1.9922632
+AGS     = 0.7096599      B0      = 1.892162E-6    B1      = 5E-6
+KETA    = 0.05           A1      = 7.799989E-4    A2      = 0.3
+RDSW    = 150            PRWG    = 0.3506787      PRWB    = 0.1097886
+WR      = 1              WINT    = 7.71429E-9     LINT    = 1.039368E-8
+DWG     = 1.205322E-8    DWB     = 8.815731E-9    VOFF    = -0.0331648
+NFACTOR = 2.5            CIT     = 0              CDSC    = 2.4E-4
+CDSCD   = 0              CDSCB   = 0              ETA0    = 2.754257E-6
+ETAB    = -0.0108095     DSUB    = 4.0643E-6      PCLM    = 1.9769478
+PDIBLC1 = 0.9710894      PDIBLC2 = 0.01           PDIBLCB = 0.1
+DROUT   = 0.9993653      PSCBE1  = 7.973288E10    PSCBE2  = 5.02618E-10
+PVAG    = 0.536394       DELTA   = 0.01           RSH     = 6.7
+MOBMOD  = 1              PRT     = 0              UTE     = -1.5
+KT1     = -0.11          KT1L    = 0              KT2     = 0.022
+UA1     = 4.31E-9        UB1     = -7.61E-18      UC1     = -5.6E-11
+AT      = 3.3E4          WL      = 0              WLN     = 1
+WW      = 0              WWN     = 1              WWL     = 0
+LL      = 0              LLN     = 1              LW      = 0
+LWN     = 1              LWL     = 0              CAPMOD  = 2
+XPART   = 0.5            CGDO    = 3.74E-10       CGSO    = 3.74E-10
+CGBO    = 1E-12          CJ      = 9.581316E-4    PB      = 0.9759771
+MJ      = 0.404514       CJSW    = 1E-10          PBSW    = 0.8002028
+MJSW    = 0.6            CJSWG   = 3.3E-10        PBSWG   = 0.8002028
+MJSWG   = 0.6            CF      = 0              PVTH0   = 2.009264E-4
+PRDSW   = 0              PK2     = 1.30501E-3     WKETA   = 7.565815E-3
+LKETA   = 0.0327047      PU0     = 4.4729531      PUA     = 1.66833E-11
+PUB     = 0              PVSAT   = 653.2294237    PETA0   = 1E-4
+PKETA   = -0.0101124      )
.MODEL CMOSP PMOS (                             
+VERSION = 3.1            LEVEL=49        TNOM    = 27             TOX     = 3.1E-9
+XJ      = 1E-7           NCH     = 4.1589E17      VTH0    = -0.2156906
+K1      = 0.2680989      K2      = 4.539197E-3    K3      = 0.097375
+K3B     = 6.5043674      W0      = 1E-6           NLX     = 2.836757E-7
+DVT0W   = 0              DVT1W   = 0              DVT2W   = 0
+DVT0    = 0              DVT1    = 1              DVT2    = 0.1
+U0      = 106.670318     UA      = 1.152986E-9    UB      = 2.377339E-21
+UC      = -1.93766E-11   VSAT    = 1.190739E5     A0      = 1.7356069
+AGS     = 0.6166218      B0      = 7.467707E-6    B1      = 4.992767E-6
+KETA    = 0.0157125      A1      = 8.723417E-3    A2      = 0.8713799
+RDSW    = 105            PRWG    = -0.5           PRWB    = 0.5
+WR      = 1              WINT    = 0              LINT    = 1.495916E-8
+DWG     = 6.058254E-9    DWB     = -1.83713E-8    VOFF    = -0.1022829
+NFACTOR = 1.5332272      CIT     = 0              CDSC    = 2.4E-4
+CDSCD   = 0              CDSCB   = 0              ETA0    = 0.0110506
+ETAB    = -2.941775E-3   DSUB    = 2.419246E-3    PCLM    = 0.2085802
+PDIBLC1 = 9.972716E-4    PDIBLC2 = -1.39497E-13   PDIBLCB = -1E-3
+DROUT   = 0.6860806      PSCBE1  = 1.849353E9     PSCBE2  = 5.675435E-10
+PVAG    = 0.0149584      DELTA   = 0.01           RSH     = 6.6
+MOBMOD  = 1              PRT     = 0              UTE     = -1.5
+KT1     = -0.11          KT1L    = 0              KT2     = 0.022
+UA1     = 4.31E-9        UB1     = -7.61E-18      UC1     = -5.6E-11
+AT      = 3.3E4          WL      = 0              WLN     = 1
+WW      = 0              WWN     = 1              WWL     = 0
+LL      = 0              LLN     = 1              LW      = 0
+LWN     = 1              LWL     = 0              CAPMOD  = 2
+XPART   = 0.5            CGDO    = 3.42E-10       CGSO    = 3.42E-10
+CGBO    = 1E-12          CJ      = 1.15643E-3     PB      = 0.8
+MJ      = 0.4399866      CJSW    = 1.133806E-10   PBSW    = 0.8
+MJSW    = 0.1146401      CJSWG   = 4.22E-10       PBSWG   = 0.8
+MJSWG   = 0.1146401      CF      = 0              PVTH0   = 1.282832E-3
+PRDSW   = 44.1361752     PK2     = 2.459655E-3    WKETA   = 0.0352131
+LKETA   = 0.0128331      PU0     = -1.2608844     PUA     = -4.27994E-11
+PUB     = 1.628153E-28   PVSAT   = -50            PETA0   = 7.039749E-5
+PKETA   = -5.052402E-3    )
.SUBCKT INVERT 1 2 3 32
 mp1 4 1 3 3 cmosp W=0.480000U l=0.13u
+AS=0.0384p AD=0.0384P PD=1.12U PS=1.12U
mn1 6 1 32 0 cmosn W=0.16U L=0.13U
+AS=0.0128P AD=0.0128P PD=0.48U PS=0.48U
r2  4 2 6
r3  2 6 7
c3  4 0 0.0000842pF
c4  6 0 0.0000815pF
.ENDS
.OP
.tran .1u 200u 0
*switching energy pmos
.meas tran switch_qp integ I(v1) from=121ns to=240ns
.meas tran switch_energyp param='5*switch_qp'


*switching energy  nmos
.meas tran switch_qn integ I(v2) from=121ns to=240ns
.meas tran switch_energyn param='5*switch_qn'
*avg leakage current
.meas tran leak_I avg I(v1) from=1ns to=600ns
*leakage energy
.meas tran leak_q integ I(v1) from=121ns to=240ns
.meas tran leakage_energy param='5*leak_q'
*leakage power
.meas tran leak_power param='5*leak_I'
.print v1
.PROBE 
.END
To run hspice, type hpice on the terminal and give the path of the .sp file.
Leakge power and Switching power measured in the critical path will be shown in the .mto file generated after running hpice, shown below:


$DATA1 SOURCE='HSPICE' VERSION='D-2010.03 32-BIT'
.TITLE 'inverter netlist'
 switch_qp        switch_energyp   switch_qn        switch_energyn    
 leak_i           leak_q           leakage_energy   leak_power        
 temper           alter#            
  4.198e-15        2.099e-14        5.038e-17        2.519e-16        
 -9.005e-09        4.198e-15        2.099e-14       -4.502e-08        
   25.0000           1.0000         

